6.001 Recitation 3:
Substitution Model, Recursion, and Iteration, Oh My!

RI: Gerald Dalley
14 Feb 2007

The Substitution Model

Newtonian physics, quantum mechanics, relativity, and Scheme...

Rules of the substitution model:

1. if **self-evaluating**, (*e.g.* number, string, `#t`, `#f`), just return value.
2. if **name**, replace it with value associated with that name.
3. if **special form** (*e.g.* `if`, `define`, `lambda`), follow the special form's rules for evaluation
   (a) if **lambda**, create a procedure
   (b) ...
4. if **combination** (e_0 e_1 ... e_n),
   (a) Evaluate subexpressions e_i in any order to produce values (v_0 v_1 ... v_n)
   (b) If v_0 is **primitive procedure** (*e.g.* `+`), just apply it to v_1 ... v_n
   (c) if v_0 is **compound procedure** (created by `lambda`):
      i. Substitute v_1 ... v_n for corresponding parameters in body of procedure, then repeat on body

Hints:
- Do one thing at a time
- Take small steps
- Sometimes we have a choice of what to do next

```
; Evaluate the following expression, one
; step at a time using the substitution model:
(if (= (+ 5 2) 7) (* (+ 2 3) 5) (/ 4 (- 7 5)))
(if (= 7       7) (* (+ 2 3) 5) (/ 4 (- 7 5)))
(if #t            (* (+ 2 3) 5) (/ 4 (- 7 5)))
(if #t            (* 5       5) (/ 4 (- 7 5)))
(if #t            25            (/ 4 (- 7 5)))
25
```

Recursive versus Iterative

From Webster:
- *recur:* to come up again for consideration
- *recursion:* something that calls itself
- *iterate:* repeating, each time getting closer to the desired result

What makes a procedure recursive?

it calls itself (possibly indirectly)

What makes a process recursive?

deferred operations

Designing Recursive Algorithms

- wishful thinking
- decompose the problem
- identify the simple/smallest parts (modular programming)

```
(define (odd? x)
  (not (even? x)))

(define (even? x)
  (not (odd? x)))

; fix even? ...
(define (even? x)
  (if (= x 0) #t (odd? (- x 1))))
```

- Enumerate assumptions
- Test/predicate for base case (termination of recursive unwinding)
- Recursive call

```
(define (fact n)
   (* n (fact (- n 1))))

; Fix fact, and list the input assumptions you make...

; Assumes: (>= n 1)
(define (fact n)
   (if (= n 1)
       1
       (* n (fact (- n 1)))))
```

High-Level Questions

What is "recursive recursion?"

- the width of the expression (substitution model) keeps growing
- deferred operations

What is "iterative recursion?"

- constant space
- no pending/deferred operations

Why do we like recursive procedures?

many problems are easy to think about as loops

Why do some programming guides discourage writing recursive programs?

- limited stack depth
- limited memory
- encourage the use of for loops

Recursive → Iterative Recipe

1. What can we use as our partial answer
   - can we "accumulate" something?
2. How do we keep track of what's left to do

- some sort of counter (count up, count down?)

3. How do we update these variables
   - the partial answer & counter
4. What's the base case?
5. Write out a table
   - fixed size (# of variables) because there are no pending operations
6. Almost always we will write a helper procedure that is the recursive part because we have extra variables to keep track of!

`biggie-size` Returns!

We encode an order as multi-digit number where each digit represents a combo. For example, the order 327 represents a Triple, Double, and biggie-sized Triple. (biggie-size = regular+4)

```
(define order-price
  (lambda (order)
    (if (= order 0)
        0
        (+ (combo-price (remainder order 10))
           (order-price (quotient order 10))))))
```

Is this recursive or iterative?

It generates a recursive process.

Rewrite as the other type:

```
; The iterative version:
(define order-price
  (lambda (order)
    (define helper
      (lambda (subtotal remaining)
        (if (= remaining 0)
            subtotal
            (helper (+ subtotal (combo-price (remainder remaining 10)))
                    (quotient remaining 10)))))
    (helper 0 order)))
```

`begin` Special Form

`(begin expr1 expr2 ...)`: The expressions are evaluated sequentially from left to right, and the value of the last expression is returned. This expression type is used to sequence side effects such as input and output.

- allows multiple statements
- scoping
- implicit `begin` inside `define`, `cond`, `lambda`, ...

Consider:

```
(define (foo a)
  (begin
    (define b 5)
    (define (foo-helper x) (+ a x))
    (foo-helper b)))
```

To what do the following expressions evaluate?

`foo` →	#[procedure]	`foo-helper` →	ERROR, undefined variable
`a` →	ERROR, undefined variable	`b` →	ERROR, undefined variable
`(foo 3)` →	8	`foo-helper` →	ERROR, undefined variable
`a` →	ERROR, undefined variable	`b` →	ERROR, undefined variable

Order of Evaluation

Consider the following code:

```
(define (our-display x)
  (display x) ; returns something wierd
  x) ; we return something sensible instead

(define (count1 x)
  (cond ((= x 0) 0)
        (else (our-display x)
              (count1 (- x 1)))))
(define (count2 x)
  (cond ((= x 0) 0)
        (else (count2 (- x 1))
              (our-display x))))
```

```
; What does the following generate?
(count1 5)
=> 543210
```

```
; What does the following generate?
(count2 4)
=> 123455
```

Evaluate using the substitution model:

```
(count1 5)
(begin (our-display 5) (count1 4)) <PRINT 5>
(count1 4)
(begin (our-display 4) (count1 3)) <PRINT 4>
(count1 3)
(begin (our-display 3) (count1 2)) <PRINT 3>
(count1 2)
(begin (our-display 2) (count1 1)) <PRINT 2>
(count1 1)
(begin (our-display 1) (count1 0)) <PRINT 1>
(count1 0)
=> 0
```

```
(count2 5)
(begin (count2 4) (our-display 5))
(begin (begin (count2 3) (our-display 4)) (our-display 5))
(begin (begin (begin (count2 2) (our-display 3)) (our-display 4)) (our-display 5))
(begin (begin (begin (begin (count2 1) (our-display 2)) (our-display 3)) (our-display 4)) (our-display 5))
(begin (begin (begin (begin (begin (count2 0) (our-display 1)) (our-display 2)) (our-display 3)) (our-displ
(begin (begin (begin (begin (our-display 1) (our-display 2)) (our-display 3)) (our-display 4)) (our-display
(begin (begin (begin (our-display 2) (our-display 3)) (our-display 4)) (our-display 5)) <PRINT 2>
(begin (begin (our-display 3) (our-display 4)) (our-display 5)) <PRINT 3>
(begin (our-display 4) (our-display 5)) <PRINT 4>
(our-display 5) <PRINT 5>
=> 5
```